
NAG Library Routine Document

C02AGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms
and other implementation-dependent details.

1 Purpose

C02AGF finds all the roots of a real polynomial equation, using a variant of Laguerre's method.

2 Specification

SUBROUTINE C02AGF (A, N, SCAL, Z, W, IFAIL)

INTEGER N, IFAIL
REAL (KIND=nag_wp) A(N+1), Z(2,N), W(2*(N+1))
LOGICAL SCAL

3 Description

C02AGF attempts to find all the roots of the nth degree real polynomial equation

P zð Þ ¼ a0z
n þ a1z

n�1 þ a2z
n�2 þ � � � þ an�1zþ an ¼ 0:

The roots are located using a modified form of Laguerre's method, originally proposed by Smith (1967).

The method of Laguerre (see Wilkinson (1965)) can be described by the iterative scheme

L zkð Þ ¼ zkþ1 � zk ¼ �nP zkð Þ
P 0 zkð Þ � ffiffiffiffiffiffiffiffiffiffiffiffiffi

H zkð Þp ;

where H zkð Þ ¼ n� 1ð Þ n� 1ð Þ P 0 zkð Þð Þ2 � nP zkð ÞP 00 zkð Þ
h i

and z0 is specified.

The sign in the denominator is chosen so that the modulus of the Laguerre step at zk, viz. L zkð Þj j, is as
small as possible. The method can be shown to be cubically convergent for isolated roots (real or
complex) and linearly convergent for multiple roots.

The routine generates a sequence of iterates z1; z2; z3; . . . , such that P zkþ1ð Þj j < P zkð Þj j and ensures that
zkþ1 þ L zkþ1ð Þ ‘roughly’ lies inside a circular region of radius Fj j about zk known to contain a zero of
P zð Þ; that is, L zkþ1ð Þj j � Fj j, where F denotes the Fejér bound (see Marden (1966)) at the point zk.
Following Smith (1967), F is taken to be min B; 1:445nRð Þ, where B is an upper bound for the
magnitude of the smallest zero given by

B ¼ 1:0001�min
ffiffiffi
n

p
L zkð Þ; r1j j; an=a0j j1=n

� �
;

r1 is the zero X of smaller magnitude of the quadratic equation

P 00 zkð Þ
2n n� 1ð ÞX

2 þ P 0 zkð Þ
n

X þ 1
2P zkð Þ ¼ 0

and the Cauchy lower bound R for the smallest zero is computed (using Newton's Method) as the
positive root of the polynomial equation

a0j jzn þ a1j jzn�1 þ a2j jzn�2 þ � � � þ an�1j jz� anj j ¼ 0:

Starting from the origin, successive iterates are generated according to the rule zkþ1 ¼ zk þ L zkð Þ, for
k ¼ 1; 2; 3; . . . , and L zkð Þ is ‘adjusted’ so that P zkþ1ð Þj j < P zkð Þj j and L zkþ1ð Þj j � Fj j. The iterative
procedure terminates if P zkþ1ð Þ is smaller in absolute value than the bound on the rounding error in
P zkþ1ð Þ and the current iterate zp ¼ zkþ1 is taken to be a zero of P zð Þ (as is its conjugate �zp if zp is
complex). The deflated polynomial ~P zð Þ ¼ P zð Þ= z� zp

� �
of degree n� 1 if zp is real

( ~P zð Þ ¼ P zð Þ= z� zp
� �

z� �zp
� �� �

of degree n� 2 if zp is complex) is then formed, and the above
procedure is repeated on the deflated polynomial until n < 3, whereupon the remaining roots are
obtained via the ‘standard’ closed formulae for a linear (n ¼ 1) or quadratic (n ¼ 2) equation.

Note that C02AJF, C02AKF and C02ALF can be used to obtain the roots of a quadratic, cubic (n ¼ 3)
and quartic (n ¼ 4) polynomial, respectively.

4 References

Providence, RI

Smith B T (1967) ZERPOL: a zero finding algorithm for polynomials using Laguerre's method

Thompson K W (1991) Error analysis for polynomial solvers Fortran Journal (Volume 3) 3 10–13

Wilkinson J H (1965) The Algebraic Eigenvalue Problem Oxford University Press, Oxford

5 Arguments

1: AðNþ 1Þ – REAL (KIND=nag_wp) array Input

On entry: if A is declared with bounds 0 : Nð Þ, then AðiÞ must contain ai (i.e., the coefficient of
zn�i) , for i ¼ 0; 1; . . . ; n.

Constraint: Að0Þ 6¼ 0:0.

2: N – INTEGER Input

On entry: n, the degree of the polynomial.

Constraint: N � 1.

3: SCAL – LOGICAL Input

On entry: indicates whether or not the polynomial is to be scaled. See Section 9 for advice on
when it may be preferable to set SCAL ¼ :FALSE: and for a description of the scaling strategy.

Suggested value: SCAL ¼ :TRUE:.

4: Zð2;NÞ – REAL (KIND=nag_wp) array Output

On exit: the real and imaginary parts of the roots are stored in Zð1; iÞ and Zð2; iÞ respectively, for
i ¼ 1; 2; . . . ; n. Complex conjugate pairs of roots are stored in consecutive pairs of elements of Z;
that is, Zð1; i þ 1Þ ¼ Zð1; iÞ and Zð2; iþ 1Þ ¼ �Zð2; iÞ.

5: Wð2� Nþ 1ð ÞÞ – REAL (KIND=nag_wp) array Workspace

6: IFAIL – INTEGER Input/Output

On entry: IFAIL must be set to 0, �1 or 1. If you are unfamiliar with this argument you should
refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is
detected, the value �1 or 1 is recommended. If the output of error messages is undesirable, then
the value 1 is recommended. Otherwise, if you are not familiar with this argument, the
recommended value is 0. When the value �1 or 1 is used it is essential to test the value of
IFAIL on exit.

On exit: IFAIL ¼ 0 unless the routine detects an error or a warning has been flagged (see
Section 6).

6 Error Indicators and Warnings

If on entry IFAIL ¼ 0 or �1, explanatory error messages are output on the current error message unit
(as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL ¼ 1

On entry, Að0Þ ¼ 0:0,
or N < 1.

IFAIL ¼ 2

The iterative procedure has failed to converge. This error is very unlikely to occur. If it does,
please contact NAG, as some basic assumption for the arithmetic has been violated. See also
Section 9.

IFAIL ¼ 3

Either overflow or underflow prevents the evaluation of P zð Þ near some of its zeros. This error is
very unlikely to occur. If it does, please contact NAG. See also Section 9.

IFAIL ¼ �99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL ¼ �999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

All roots are evaluated as accurately as possible, but because of the inherent nature of the problem
complete accuracy cannot be guaranteed. See also Section 10.

8 Parallelism and Performance

C02AGF is not threaded in any implementation.

9 Further Comments

If SCAL ¼ :TRUE:, then a scaling factor for the coefficients is chosen as a power of the base b of the
machine so that the largest coefficient in magnitude approaches thresh ¼ bemax�p. You should note that
no scaling is performed if the largest coefficient in magnitude exceeds thresh, even if SCAL ¼ :TRUE:.
(b, emax and p are defined in Chapter X02.)

However, with SCAL ¼ :TRUE:, overflow may be encountered when the input coefficients
a0; a1; a2; . . . ; an vary widely in magnitude, particularly on those machines for which b 4pð Þ overflows.
In such cases, SCAL should be set to .FALSE. and the coefficients scaled so that the largest coefficient
in magnitude does not exceed b emax�2pð Þ.

Even so, the scaling strategy used by C02AGF is sometimes insufficient to avoid overflow and/or
underflow conditions. In such cases, you are recommended to scale the independent variable zð Þ so that

the disparity between the largest and smallest coefficient in magnitude is reduced. That is, use the
routine to locate the zeros of the polynomial dP czð Þ for some suitable values of c and d. For example,
if the original polynomial was P zð Þ ¼ 2�100 þ 2100z20, then choosing c ¼ 2�10 and d ¼ 2100, for
instance, would yield the scaled polynomial 1þ z20, which is well-behaved relative to overflow and
underflow and has zeros which are 210 times those of P zð Þ.
If the routine fails with IFAIL ¼ 2 or 3, then the real and imaginary parts of any roots obtained before
the failure occurred are stored in Z in the reverse order in which they were found. Let nR denote the
number of roots found before the failure occurred. Then Zð1; nÞ and Zð2; nÞ contain the real and
imaginary parts of the first root found, Zð1; n� 1Þ and Zð2; n� 1Þ contain the real and imaginary parts
of the second root found, . . . ;Zð1; n� nR þ 1Þ and Zð2; n� nR þ 1Þ contain the real and imaginary
parts of the nRth root found. After the failure has occurred, the remaining 2� n� nRð Þ elements of Z
contain a large negative number (equal to �1= X02AMFðÞ � ffiffiffi

2
p� �

).

10 Example

For this routine two examples are presented. There is a single example program for C02AGF, with a
main program and the code to solve the two example problems given in the subroutines EX1 and EX2.

Example 1 (EX1)

This example finds the roots of the fifth degree polynomial

z5 þ 2z4 þ 3z3 þ 4z2 þ 5zþ 6 ¼ 0:

Example 2 (EX2)

This example solves the same problem as subroutine EX1, but in addition attempts to estimate the
accuracy of the computed roots using a perturbation analysis. Further details can be found in Thompson
(1991).

10.1 Program Text

! C02AGF Example Program Text
! Mark 26 Release. NAG Copyright 2016.

Module c02agfe_mod

! C02AGF Example Program Module:
! Parameters

! .. Implicit None Statement ..
Implicit None

! .. Accessibility Statements ..
Private

! .. Parameters ..
Integer, Parameter, Public :: nin = 5, nout = 6
Logical, Parameter, Public :: scal = .True.

End Module c02agfe_mod
Program c02agfe

! C02AGF Example Main Program

! .. Use Statements ..
Use c02agfe_mod, Only: nout

! .. Implicit None Statement ..
Implicit None

! .. Executable Statements ..
Write (nout,*) ’C02AGF Example Program Results’

Call ex1

Call ex2

Contains
Subroutine ex1

! .. Use Statements ..
Use nag_library, Only: c02agf, nag_wp
Use c02agfe_mod, Only: nin, scal

! .. Local Scalars ..
Real (Kind=nag_wp) :: zi, zr
Integer :: i, ifail, n, nroot

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:), w(:), z(:,:)

! .. Intrinsic Procedures ..
Intrinsic :: abs

! .. Executable Statements ..
Write (nout,*)
Write (nout,*)
Write (nout,*) ’Example 1’

! Skip heading in data file
Read (nin,*)
Read (nin,*)
Read (nin,*)

Read (nin,*) n
Allocate (a(0:n),w(2*(n+1)),z(2,n))

Read (nin,*)(a(i),i=0,n)

Write (nout,*)
Write (nout,99999) ’Degree of polynomial = ’, n

ifail = 0
Call c02agf(a,n,scal,z,w,ifail)

Write (nout,99998) ’Computed roots of polynomial’

nroot = 1

Do While (nroot<=n)

zr = z(1,nroot)
zi = z(2,nroot)
If (zi==0.0E0_nag_wp) Then

Write (nout,99997) ’z = ’, zr
nroot = nroot + 1

Else
Write (nout,99997) ’z = ’, zr, ’ +/- ’, abs(zi), ’*i’
nroot = nroot + 2

End If

End Do

99999 Format (/,1X,A,I4)
99998 Format (/,1X,A,/)
99997 Format (1X,A,1P,E12.4,A,E12.4,A)

End Subroutine ex1
Subroutine ex2

! .. Use Statements ..
Use nag_library, Only: a02abf, c02agf, nag_wp, x02ajf, x02alf
Use c02agfe_mod, Only: nin, scal

! .. Local Scalars ..
Real (Kind=nag_wp) :: deltac, deltai, di, eps, epsbar, f, &

r1, r2, r3, rmax
Integer :: i, ifail, j, jmin, n

! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:), abar(:), r(:), w(:), z(:,:), &

zbar(:,:)
Integer, Allocatable :: m(:)

! .. Intrinsic Procedures ..
Intrinsic :: abs, max, min

! .. Executable Statements ..
Write (nout,*)
Write (nout,*)

Write (nout,*) ’Example 2’

! Skip heading in data file
Read (nin,*)
Read (nin,*)

Read (nin,*) n
Allocate (a(0:n),abar(0:n),r(n),w(2*(n+1)),z(2,n),zbar(2,n),m(n))

! Read in the coefficients of the original polynomial.

Read (nin,*)(a(i),i=0,n)

! Compute the roots of the original polynomial.

ifail = 0
Call c02agf(a,n,scal,z,w,ifail)

! Form the coefficients of the perturbed polynomial.

eps = x02ajf()
epsbar = 3.0_nag_wp*eps

Do i = 0, n

If (a(i)/=0.0_nag_wp) Then
f = 1.0_nag_wp + epsbar
epsbar = -epsbar
abar(i) = f*a(i)

Else
abar(i) = 0.0E0_nag_wp

End If

End Do

! Compute the roots of the perturbed polynomial.

ifail = 0
Call c02agf(abar,n,scal,zbar,w,ifail)

! Perform error analysis.

! Initialize markers to 0 (unmarked).

m(1:n) = 0

rmax = x02alf()

! Loop over all unperturbed roots (stored in Z).

Do i = 1, n
deltai = rmax
r1 = a02abf(z(1,i),z(2,i))

! Loop over all perturbed roots (stored in ZBAR).

Do j = 1, n

! Compare the current unperturbed root to all unmarked
! perturbed roots.

If (m(j)==0) Then
r2 = a02abf(zbar(1,j),zbar(2,j))
deltac = abs(r1-r2)

If (deltac<deltai) Then
deltai = deltac
jmin = j

End If

End If

End Do

! Mark the selected perturbed root.

m(jmin) = 1

! Compute the relative error.

If (r1/=0.0E0_nag_wp) Then
r3 = a02abf(zbar(1,jmin),zbar(2,jmin))
di = min(r1,r3)
r(i) = max(deltai/max(di,deltai/rmax),eps)

Else
r(i) = 0.0_nag_wp

End If

End Do

Write (nout,*)
Write (nout,99999) ’Degree of polynomial = ’, n
Write (nout,*)
Write (nout,*) ’Computed roots of polynomial ’, ’ Error estimates’
Write (nout,*) ’ ’, &

’ (machine-dependent)’
Write (nout,*)

Do i = 1, n
Write (nout,99998) ’z = ’, z(1,i), z(2,i), ’*i’, r(i)

End Do

99999 Format (1X,A,I4)
99998 Format (1X,A,1P,E12.4,Sp,E12.4,A,5X,Ss,E9.1)

End Subroutine ex2
End Program c02agfe

10.2 Program Data

C02AGF Example Program Data

Example 1
5

1.0 2.0 3.0 4.0 5.0 6.0

Example 2
5

1.0 2.0 3.0 4.0 5.0 6.0

10.3 Program Results

C02AGF Example Program Results

Example 1

Degree of polynomial = 5

Computed roots of polynomial

z = -1.4918E+00
z = 5.5169E-01 +/- 1.2533E+00*i
z = -8.0579E-01 +/- 1.2229E+00*i

Example 2

Degree of polynomial = 5

Computed roots of polynomial Error estimates

(machine-dependent)

z = -1.4918E+00 +0.0000E+00*i 1.0E-15
z = 5.5169E-01 +1.2533E+00*i 1.1E-16
z = 5.5169E-01 -1.2533E+00*i 1.1E-16
z = -8.0579E-01 +1.2229E+00*i 1.5E-16
z = -8.0579E-01 -1.2229E+00*i 1.5E-16

